


Lean Formalization Seminar

Lean is a functional programming language with a community-developed library of formalized mathematics called *Mathlib* that is used to express mathematical theorems in a format that can be verified programmatically. The *Mathlib* library provides the semantic structure and definitions for proofs and statements, while the Lean language has a kernel that verifies that the logic is consistent. This is possible thanks to Lean’s “functional” paradigm (relating to functions), which checks that objects have self-consistent types. In this framework, the library acts as an additional layer that redefines notation to a high degree.

Contents

1. Installation and Project Setup	2
2. Type System	2
3. The Check Command	3
4. Functional Syntax	4
4.1. Namespaces	4
5. Tactics	5
6. Basic Tactics	6
6.1. Reverse Directions	8
7. The Variable Keyword	8
8. Searching for Tactics	8
9. Calc Mode	9
10. Control Flow Tactics	9
10.1. Changing Goal	10
10.2. Equivalences	12
10.3. Glimpse Of Lean - Probability	13
11. Additional References	17
11.1. Sources	17
11.2. Complete Projects	17
11.3. Smaller Projects	17



1. Installation and Project Setup

You may install Lean to code on VSCode by following [the official guide](#). Lean may also be used online, with no installation [at this link](#).

1. **Create a project:** in VS Code, click the ∇ symbol in the top right and choose **New Project**. To work on an existing project, choose **Open Project**. Depending on internet speed, setup may take some time, but VS Code shows progress while the project is being prepared.
2. **Open the folder:** use VScode **File: Open Folder**. Explore the files to get a sense of project structure, but only edit the ones in the `My_project_name/` folder, as the others are managed automatically.
3. **Import Mathlib:** you can import the mathematical library by writing `import Mathlib` as the first line in the `Basic.lean` file.
4. **Lean4 InfoView:** when opening `.lean` files, you can use **Lean 4: InfoView: Toggle InfoView** to understand the intermediate state of a proof. Placing the cursor in the middle of the proof will show the available information and the necessary steps next. If strange errors appear but the proof is correct, you can press the **Restart File** button

Formalization projects start with a blueprint, which gets filled progressively: a [sketch of the idea](#) can be found on Terence Tao's blog, while you may find here a list of [exemplars and the installation guide](#). For reference, this is an [overview of the available theorems](#),

2. Type System


LEAN's language is based on the axioms of **Dependent Type Theory**. At a very high level, this allows for the existence of functions where the value of the first parameter can affect the type, and thus the value, of the later parameters that a function takes in. In this context, there is no difference between a function in the mathematical sense and in the programming sense.

To help visualize this change, consider the difference between the standard Cartesian Product and the type-dependent product:

- *Cartesian Product:* $\{(a, b) : a \in A, \quad b \in B\}$, here we combine two sets in a completely independent way.
- *Dependent Product:* $\{(a, b) : a \in A, \quad b \in B_a\}$, in this case the value of a directly affects the type, and thus possible values of b . Note that this assumes the existence of a type family $A \rightarrow \text{Type}$. The type of the second component depends on the value of the first.

This allows for functions to behave differently based on their inputs, but also to construct high level arguments by abstracting from the details. Just like in mathematics, where the same proof can be applied to all objects that share a specific algebraic structure, we can imagine a proof like a function that first takes as an argument the type of the object, followed by some fixed assumptions on the objects, which depend on the object, but may be imposed for all objects in a more general collection.

The second peculiarity of Lean's type system is the existence of an infinite hierarchy of universes. Propositions live in **Prop**, while data types live in **Type u** for some universe level **u**. In particular, **Type 0 : Type 1**, **Type 1 : Type 2**, and so on. This hierarchy prevents inconsistencies such as **Type : Type** and helps avoid paradoxes related to self-reference, such as Russell's paradox in naive set theory. Intuitively, collections of objects from one universe live in a higher universe, so a type is not itself an element of the same




type. In practice, Lean infers universe levels automatically, assigning constructions to higher universes when needed.

For instance, consider the notion of a linear function. Since linearity can be defined for a wide class of objects, the same proof may be used on different spaces, by simply fixing the overarching space, and then defining a function on this space. This would be an example of type dependence.

For Mathlib formalizations, we must consider two fundamental objects:

- **Proposition:** object that expresses a statement, without imposing any assumption of it being true or false. They are constructed by combining basic facts through logic. For example, the statement `n = 2 → Even n` is itself a proposition. To help visualize this, consider the statement `n=2 => n even`, which is built by mapping statements `n=2` and `n even` through an implication function taking the form `imply : Statement x Statement -> Statement`. Since false statements may also be constructed in this way, thus we need another object to verify a statement.
- **Proof:** a term whose type is the proposition being proved. It is an object that represents a series of steps that results in the construction of a Proposition. In Mathlib, a proof is an object of type proposition, and it is at the same time a function that “transforms” some assumptions into a series of conclusions. Following the previous example, a proof that `n=2` implies `n even` would be an object of type “Statement that `n=2 => n even`”, but it also may be equivalently formulated as a function that given an object of type `n=2` (that is a proof of the statement), constructs an object of type `n even`.



Proof Irrelevance: when verifying the correctness of a statement, the Lean language verifies that each intermediate step follows from the previous one, by simply checking that the types of the proofs are always compatible. In this fashion, all proofs of a statement are equivalent for the sake of verifying the statement.

Following this formalism, **axioms** act simply as an object that stands in for a proof, but whose definition is not explicitly stated, but is taken as a given. In fact, axioms take the role of lemmas during the intermediate steps in the logical chain of a proof. A similar notion is true in the case of hypotheses: any assumptions in the statement are equivalent in role to having a proof of that hypothesis. In brief, axioms and hypotheses provide terms that may be used as proofs of the corresponding propositions. This is natural, since for both of these statements we can use the fact they are an axiom or a hypothesis as a reason enough to treat them as true.

Another useful insight to this philosophy is in the case of proofs of implication. Suppose you have a proof that $a \wedge b \Rightarrow b \wedge c$. This is an inhabitant of the set $a \wedge b \Rightarrow b \wedge c$, but it can also be viewed as a function that converts a proof of statement $a \wedge b$ into a proof of type $b \wedge c$. In a sense, proofs of implication are bridges between different sets of proofs, whether they are assumptions, axioms or proofs that you have implemented yourself.

3. The Check Command

The exact definition of types may sometimes get confusing, since objects in Lean are capable of taking other objects and perform meta-programming changes on them. To navigate these issues, `#check obj` can be used to verify or infer types. This helps understand the language better, by identifying the

```
variable (a b c : ℝ)

#check a
#check a + b
#check (a : ℝ)
#check mul_comm a b
```

meaning of a term, or to figure out what term is being introduced in a proof. In a similar fashion it is also possible to jump to the definition of a term in code with the shortcut `ctrl/cmd + click`.

```
#check mul_comm
#check (Type 20 → Type 21)
```

The check command may be used anywhere, even in the middle of the statement of a proof.

4. Functional Syntax

The syntax of Mathlib is simple, but all-encompassing by using **syntax overloading**: unicode characters like $(\forall, \exists, \mathbb{N}, \mathbb{Z}, \mathbb{Q}, \mathbb{R}, \leq, \neq, \in, \sum, \prod, \int \dots)$ are used to represent complex concepts concisely. Most symbols can be inserted by simply writing “\” and the character shorthand, but more can be defined as needed. Moreover, hovering the mouse over a symbol will show the proper shorthand.

The fundamental unit of knowledge in Mathlib is that values have types. This is expressed using the `value : type` notation, which is used to construct all other complex functions. Consider the following definition of a “function”:

```
theorem Complex.cos_add_sin_mul_I_pow (n : ℕ) (z : ℂ) :
  (cos z + sin z * I) ^ n = cos (↑n * z) + sin (↑n * z) * I
```

Here, the **theorem** keyword explains the definition, the name is `Complex.cos_add_sin_mul_I_pow`, and the type of the arguments is one variable `n` of type natural number and one variable `z` of type complex number, while the type of the output of the function is the statement of *De Moivre’s theorem*. Observe also the presence of the `↑` character, which communicates upcasting: we convert the natural number `n` into a complex number, so that multiplication with a complex number is defined.

You may also encounter curly bracket arguments `{...}`, which fix the default values for the arguments, allowing them to be omitted, and square bracket arguments `[...]`, which tell Lean to construct a type with some additional properties on the spot, typically for equipping parameters with some additional properties.

For example, the following is the definition of `Measurable`:

```
def Measurable [MeasurableSpace α] [MeasurableSpace β] (f : α → β) : Prop :=
  ∀ {t : Set β}, MeasurableSet t → MeasurableSet (f ⁻¹ t)
```

4.1. Namespaces

To avoid confusion for cases where the same name is used in different theories, **namespaces** are used, which consist of areas of code with a fixed set of names. To use names from a namespace without qualification, you can write `open Foo`, which affects name resolution in the current scope. By contrast, `import` loads another module and does not automatically open its namespaces. After defining a namespace, you can refer to its contents with qualified names such as `Foo.x`, open the namespace with `open Foo`, or introduce notation from the scope using `open scoped Foo`.

```
namespace Foo
-- defines a new meaning for a unicode symbol in this scope
scoped infix:68 " ≈ " => BEq.beq -- boolean equality
```

```
def x := 1 -- we can define an object in a scope

-- `x` is already accessible from outside as `Foo.x`
end Foo
```

After defining a namespace, we can access its contents by referring to the namespace, importing it, or just importing the notation.

```
-- we refer to them differently based on access rule
#check Foo.x

-- but we can also import the whole namespace
open Foo
-- now we do not need the "Foo." prefix
#check x

-- or just import the syntax rules (such as for unicode symbols)
open scoped Foo
#check (1 ≈ 2)
#check (· ≈ ·)
```

Namespaces become relevant when writing larger projects. We introduce them early to notify that they exist, but they will not be useful for some time.

5. Tactics

In order to lighten the burden of proving something one step at a time, Mathlib uses **meta-programming** extensively: there are algorithms that can quickly prove simple statements by taking the current state of the proof and attempting to construct a goal, which also gets indicated. This is very common with algebraic manipulations like equalities and inequalities, for which powerful tactics exist for rearranging terms within an expression. Tactics are introduced through the `by` keyword in the following way.

Some fundamental tactics, picked from the [longer cheatsheet](#):

- `intro`: If the goal is $\forall x, P x$ or $A \rightarrow B$, introduces a new variable/hypothesis into the context (`intro x`, `intro h`), leaving the body as the new goal; supports patterns like `intro (a, b)`.
- `unfold`: Expands a definition by replacing it with its body (`unfold f`, `unfold f g`); can target locations (`unfold f at h`).
- `rw`: Rewrites occurrences using equalities/iff (`rw [h]` left-to-right, `rw [← h]` right-to-left); can rewrite in hypotheses (`rw [h] at hx`) and chain rules (`rw [h1, h2]`).
- `ring`: Closes goals of the form `lhs = rhs` in commutative (semi)rings by normalizing both sides as polynomials in the ring operations; `ring!` is a more aggressive variant.
- `rel`: For relational goals like $\leq / <$, transforms the goal using monotonicity/congruence lemmas and the inequalities you provide; e.g. `rel [h1, h2]` reduces the goal to simpler subgoals.
- `have`: Adds a new local fact to context (`have h : P := t` or `have h : P := by ...`); supports destructuring patterns like `have (h1, h2) := h`.
- `exact`: Closes the goal by providing a term whose type is definitionally equal to the goal (`exact t`).
- `simp`: Simplifies the goal (or hypotheses) using simp lemmas and definitional reductions; common forms: `simp`, `simp [h1, h2]`, `simp at h`, `simp [*]`, `simp only [lemmas...]`.

- `linarith`: Solves goals that are linear equalities/inequalities by combining linear (in)equalities from the context (derives contradictions and bounds automatically).

6. Basic Tactics

We first consider an overview of the basic tactics which can be used to prove standard arithmetic statements or algebraic manipulations

rewrite

We can use `rw` to substitute the hypothesis into an equation. In the following example, we show a trivial implication of the form `a => b` by simply substituting in hypothesis.

```
example (a b : ℝ) (h : a = b) : a = b := by
  rw [h]
```

It is also possible to use lemmas directly, helping `ring` explore the correct direction. For example, in the following example, we can tell Mathlib to use the transformation $b * a = a * b$, using the `mul_comm` lemma.

```
example (a b c : ℝ) : a * b * c = b * (a * c) := by
  rw [mul_comm a b] -- first obtain that b * a = a * b from a * b
  rw [mul_assoc b a c] -- apply associativity as (a * b) * c = b * (a * c)
```

Notably, we may also omit terms that are clear from context. This notational compactness is possible thanks to Mathlib's efficient type checking system, which connects the dots by looking at what types (statement proofs) are available and which are needed.

```
example (a b c : ℝ) : a * b * c = b * c * a := by
  rw [mul_assoc] -- apply substitution, but with the inverted hypothesis
  rw [mul_comm] -- apply substitution, but with the inverted hypothesis
```

For completeness, the definition of `mul_comm` is the following. It takes two elements of a Commutative [Magma](#) and produces the commute statement. Note how `u_1`, the universe of the magma's support (in the sense of hierarchy of types), and `G` are both inferred from context, which is why we do not provide them when invoking the statement.

```
mul_comm.{u_1} {G : Type u_1} [CommMagma G] (a b : G) : a * b = b * a
```

ring

The `ring` tactic helps to simplify an expression by performing simple operations like cancelling terms. It works by rearranging both sides of an expression to canonical normal form and verifying that they are equal.

```
example : (a + b) * (a - b) = a ^ 2 - b ^ 2 := by
  ring -- can be simplified directly

example (a b c d : ℝ) (hyp : c = d * a + b) (hyp' : b = a * d)
  : c = 2 * a * d := by
  rw [hyp, hyp'] -- we simply apply both hypotheses
  ring -- simplify using the ring tactic
```

show

In a more nuanced way, sometimes we want to restate the current goal in a form that is easier to prove. We use `show` to replace the target by a more convenient one, which Lean accepts as the same goal.

```
example (n : Nat) : n + 0 = n := by
  show n = n
  rfl -- reflexivity property of the "equality" relation
```

The tactic may also be used to make the target explicit before giving the proof term.

```
example (a b : ℝ) : a + b = b + a := by
  show a + b = b + a -- we first rearrange the goal
  exact add_comm a b -- this is the exact end goal, by just using add_comm
```

There is also a term-style version, which is guided by lemmas:

```
example (a b c : ℝ) : a * b * c = b * (a * c) := by
  show a * b * c = b * (a * c) from by -- statement rearranging
  rw [mul_comm a b, mul_assoc] -- tactic that gets used
```

relation

We can use the `rel` tactic to apply an inequality which we have as a hypothesis to a statement. As before, this is a fundamental building block.

```
example {r s : ℚ} (h1 : s + 3 ≥ r) (h2 : s + r ≤ 3) : r ≤ 3 :=
  calc
    r = (s + r + r - s) / 2 := by ring
    _ ≤ (3 + (s + 3) - s) / 2 := by rel [h1, h2]
    _ = 3 := by ring
```

Exercises: [First](#), [Second](#), [Third](#), [Fourth](#)

Use `norm_num` and `linarith [sq_nonneg (x - y)]` in exercises 3 and 4. You can use `#check` to understand how to use them.

apply

The `apply` tactic can be used to invoke a theorem in a specific setting. It also allows some generality on exact use. Also, we can use `·` to guide the subgoals, helping steer the tactic in the right direction.

```
-- transitivity of the relation "smaller or equal to"
#check (le_trans : a ≤ b → b ≤ c → a ≤ c)

example (x y z : ℝ) (h₀ : x ≤ y) (h₁ : y ≤ z) : x ≤ z := by
  apply le_trans -- steer the tactic in two steps:
  · apply h₀
  · apply h₁
```

linear arithmetic

The `linarith` tactic can handle basic linear arithmetic, such as filling in the steps required to simplify an equality or a basic inequality.

```
example (h : 2 * a ≤ 3 * b) (h' : 1 ≤ a) (h'' : d = 2) : d + a ≤ 5 * b := by
  linarith -- simple enough that it can be solved directly
```

6.1. Reverse Directions

This is not a tactic, but it is commonly used: some relations like equality may have to be applied applied in reverse order. This requires the $\leftarrow$ symbol, which is written with `\leftarrow`.

```
example : (a + b) * (a + b) = a * a + 2 * (a * b) + b * b :=
  calc
    (a + b) * (a + b) = a * a + b * a + (a * b + b * b) := by
      rw [mul_add, add_mul, add_mul]
    _ = a * a + (b * a + a * b) + b * b := by
      rw [← add_assoc, add_assoc (a * a)] -- reverse associativity
    _ = a * a + 2 * (a * b) + b * b := by
      rw [mul_comm b a, ← two_mul] -- reverse order
```

Listing 1: A more complex example using a couple of theorems

Exercises: [First](#), [Second](#)

7. The Variable Keyword

The `variable` keyword is used to define new objects, so they can be reused multiple times over different theorems or examples.

```
variable (x y z : ℝ)

-- these examples use x, y, z from the `variable` command above.

example : x + y = y + x := by
  sorry

example : (x + y) + z = x + (y + z) := by
  sorry

example : x ^ 2 >= 0 := by
  sorry

example (h : x = y) : x^2 = y^2 := by
  sorry
```

8. Searching for Tactics

To help navigate Mathlib's multitude of theorems and tactics, there are many resources that can be used to identify their exact names:

- `apply?`: the InfoView panel can suggestion an appropriate tactic.

```
example : 0 ≤ a ^ 2 := by
  apply?
```

$\Rightarrow$

Try this:
[apply] exact sq_nonneg a


- Guessing: Lean4 [naming conventions](#) help maintain consistency across libraries, allowing users to guess exact theorem names.
- Search Engines: [Lean Explore](#) and [Moogoo](#) provide semantic search capabilities.

- **Zulip**: the official Lean forum, provides users with a **Is there code for X?** [page](#).

9. Calc Mode

When we write **calc** after the **by** keyword, we are switching to a step-by-step derivation which allows us to concatenate expressions with justifications in a more readable way. Each line proves that the current expression is related to the previous one, and Lean combines the lines using transitivity. It is mainly used for equalities, inequalities, and other transitive relations

We should now have enough of an understanding to read a more complex proof, and understand it to some extent. The following example proves that the affine combination between Lotteries (a probability distribution over a countable support X) has probability that sums to 1.



```
private lemma mix_sum_one (p q : Lottery X) (α : Real) :
  -- sum over the support of X, where x is the index
  ∑ x, (α * p.val x + (1 - α) * q.val x) = 1 := by

  -- use the calc tactic to prove by direct simplification
  calc

  -- rewrite the left side, and reach the right side
  ∑ x, (α * p.val x + (1 - α) * q.val x)

  -- the new goal is written on the next line,
  -- but the justification comes from the tactic after `by`
  _ = α * ∑ x, p.val x + (1 - α) * ∑ x, q.val x := by

    -- the rewrite tactic, which uses `Finset` properties
    rw [Finset.sum_add_distrib, Finset.mul_sum, Finset.mul_sum]

  -- use rewrite again with the properties of `p` and `q` Lotteries
  _ = α * 1 + (1 - α) * 1 := by rw [p.property.2, q.property.2]

  -- use the `ring` tactic to do simple algebraic cancellations
  _ = 1 := by ring
```

Listing 2: von Neumann-Morgenstern Utility Theorem - lemma

10. Control Flow Tactics

Some tactics involve the manipulation of the proof step, being used to divide the proofs into cases, introduce intermediate steps, or conclude a proof.

exact

The **exact** tactic can be used to directly tell Lean that we have reached our conclusion. With this keyword, we explicitly communicate that we have reached our goal and the proof is complete.

```
example (P Q : Prop) (hp : P) (hq : Q) : P := by
  exact hp -- closes the goal because hp is exactly of type P

-- using exact with a function application
example (n : ℕ) (h : n = 5) : n + 0 = 5 := by
  rw [add_zero] -- simplify n + 0 to n
  exact h      -- the goal is now n = 5, which matches h exactly
```

It is also possible to “guide” the tactic by providing lemmas to it: in the following example, the lemma `sq_pos_of_pos` produces a proof that the square is also positive, from a proof that the number is positive.

```
example (a : ℝ) (ha : 0 < a) : 0 < a^2 := by
  exact sq_pos_of_pos ha
```

10.1. Changing Goal

When writing a proof, the Lean `InfoView` keeps track of all the statements that have been introduced so far and the goal that you are trying to prove. The last step connects the current state of the proof to a goal with one tactic; clearly, the proof is complete and minimal if it links every assumption to the final goal.

When constructing more complex proofs, it may be necessary to change the current goal for a variety of reasons: proving a simpler statement first, simplifying the goal or proving an equality in two steps. The following proofs are useful in such cases.

have

We can use the `have` tactic to introduce an intermediate goal, that is, an additional hypothesis, in the middle of the proof.

```
example {a b : ℝ} : 2 * a * b ≤ a^2 + b^2 := by
  have h : 0 ≤ a^2 - 2 * a * b + b^2 := -- define the new hypothesis
  calc                                     -- prove it with calc
    0 ≤ (a - b)^2 := by apply pow_two_nonneg
    _ = a^2 - 2 * a * b + b^2 := by ring

  linarith -- close the goal with linarith
```

More powerful tactics also exist, with the tradeoff that greater automation results in slower verification. For this reason, powerful tactics are often replaced with more precise solutions in the foundational theorems in Mathlib.


Exercises: [First](#), [Second](#), [Third](#), [Fourth](#)

apply

The `apply` keyword may be used to act on the goal by reducing it to a simpler form: `apply e` takes a term `e`, looks at the conclusion of its type, and tries to match that conclusion with the current goal. If it matches, Lean turns the remaining arguments of `e` into new subgoals. The following example is more involved: we first prove as a subgoal that $a^2 > 0$, which follows from the $a > 0$ assumption; we apply the same lemma, relying on this fact, completing the proof.

```
example (a : ℝ) (ha : 0 < a) : 0 < (a^2)^2 := by
  have h2 : 0 < a^2 := by -- we declare `0 < a^2` as a subgoal
    apply sq_pos_of_pos -- we start proving the subgoal
    exact ha             -- concludes the subgoal
  exact sq_pos_of_pos h2 -- we finished the subgoal, and now we prove the main
                        goal using it.
```

Exercises: [First](#)



intro

To prove implications, of the form $A \Rightarrow B$, one starts by assuming A and proceeds with obtaining the statement B . The `intro` tactic is used in these case: it takes an assumption from the goal and moves it into the local context.

```
-- Using intro to handle implications
example (P Q : Prop) : P → Q → P := by
  intro hP -- Moves 'P' into the context as hypothesis 'hP'
  intro hQ -- Moves 'Q' into the context as hypothesis 'hQ'
  exact hP -- The goal is now 'P', which we have

-- Using intro for universal quantification
example : ∀ (n : ℕ), n + 0 = n := by
  intro n -- Picks an arbitrary natural number 'n'
  rw [add_zero]
```

Another delicate example is the following, where we prove a logical implication. To read the proof, remember that in Lean, `f x` means applying `f` to `x`, and `f x y` means `(f x) y`. Also, implication is right-associative, so `p → q → r` means `p → (q → r)`. In other words, it is a function that takes a proof of `p` and returns a function that takes a proof of `q` and returns a proof of `r`.

```
example (p q r : Prop) : (p → q) → (p → q → r) → p → r := by
  -- h1 : p → q, h2 : p → q → r, h3 : p
  intro h1 h2 h3

  -- From h1 and h3, obtain a proof of q.
  have hq : q := h1 h3


  -- From h2, h3, and hq, obtain a proof of r.
  -- First, h2 h3 : q → r
  -- Then, (h2 h3) hq : r
  have hr : r := h2 h3 hq

  exact hr
```

constructor

In order to prove statements with a direct and converse proof, we can use the `constructor` keyword in the following way: we use the scope of a constructor, then prove each implication direction by first introducing the assumption with the `intro` keyword. We also use the bullet point `·` character to focus the current goal, which hides the final goal.

```
example (a b : ℝ) : (a-b)*(a+b) = 0 ↔ a^2 = b^2 := by
  constructor
  · intro h -- h : (a - b) * (a + b) = 0
    calc
      a ^ 2 = b^2 + (a - b) * (a + b) := by sorry
      _ = b^2 + 0 := by sorry
      _ = b^2 := by sorry
  · intro h -- h : a ^ 2 = b ^ 2
    calc
      (a-b)*(a+b) = a^2 - b^2 := by sorry
      _ = b^2 - b^2 := by sorry
      _ = 0 := by sorry
```



In this case, Lean helps us by dividing the proof into the two canonical directions of implication, so we can prove one at a time. At the end the `constructor` packs the two sub-proofs into a single combined structure, that is, the *iff* statement.


10.2. Equivalences

On a related note, there are simpler ways than the `constructor` tactic to prove equivalences between statements: one can rewrite an equality to obtain a trivially true equation, such as `1 = 1`.

```
-- `sub_nonneg` :  $0 \leq y - x \leftrightarrow x \leq y$ 
example {a b c : ℝ} : c + a ≤ c + b ↔ a ≤ b := by
  rw [← sub_nonneg] -- `rw` takes an equivalence
  have key : (c + b) - (c + a) = b - a := by
    ring
  rw [key] -- `rw` applies the equality `key`
  rw [sub_nonneg] -- switch back to reach the tautology  $a \leq b \leftrightarrow a \leq b$ 
```

Once you have an *iff* statement, you can invoke one of its directions of implication with the `name.1` or `name.2` syntax.

```
-- `add_le_add_iff_right` (c : ℝ) :  $a + c \leq b + c \leftrightarrow a \leq b$ 
example {a b : ℝ} (ha : 0 ≤ a) : b ≤ a + b := by
  calc
    b = 0 + b := by ring
    _ ≤ a + b := by exact (add_le_add_iff_right b).2 ha
```



10.3. Glimpse Of Lean - Probability

We can now understand a more practical series of examples.

Symmetry of Independence

```
lemma IndepSet.symm : IndepSet A B →
  IndepSet B A := by
  -- introduce the assumption
  intro h
  -- expand the definition
  unfold IndepSet
  -- swap order of two terms
  rw [inter_comm, mul_comm]
  exact h
```

$$\begin{aligned}
 B &\parallel A \\
 \mathbb{P}(B \cap A) &= \mathbb{P}(B) * \mathbb{P}(A) \\
 \mathbb{P}(A \cap B) &= \mathbb{P}(B) * \mathbb{P}(A) \\
 \mathbb{P}(A \cap B) &= \mathbb{P}(A) * \mathbb{P}(B) \\
 A &\parallel B
 \end{aligned}$$

Assume that A is independent of B . By definition, this means probability of $A \cap B$ factors as the product of the probabilities of A and B . By commutativity of set intersection and of multiplication over the reals, this is exactly our statement.

Lemmas of Complement Arguments

```
theorem Set.compl_eq_univ_diff
  (s : Set α) :
  sᶜ = univ \ s
```

```
theorem Set.inter_univ
  (s : Set α) :
  s ∩ univ = s
```

```
theorem MeasureTheory.measure_diff
  (h : s₂ ⊆ s₁)
  (h₂ : NullMeasurableSet s₂ μ)
  (h_fin : μ s₂ ≠ ⊤) :
  μ (s₁ \ s₂) = μ s₁ - μ s₂
```

```
theorem ENNReal.mul_sub
  (h : 0 < z → z < y → x ≠ ⊤) :
  x * (y - z) = x * y - x * z
```

```
theorem MeasureTheory.measure_compl
  (h₁ : MeasurableSet s)
  (h_fin : μ s ≠ ⊤) :
  μ sᶜ = μ Set.univ - μ s
```

`compl_eq_univ_diff` rewrites a complement as the difference between the whole space and the set itself, of standard use in measure identities.

`inter_univ` simplifies intersections with the universal set, such as in $\Omega \setminus A = \Omega \cap A^c = A^c$.

`measure_diff` computes the measure of a set difference $s_1 \setminus s_2$ in terms of $\mu s_1 - \mu s_2$, provided measurability, subset and finiteness assumptions.

`mul_sub` distributes multiplication across subtraction in `ENNReal`, allowing the algebraic side of the independence identity to be rearranged.

`measure_compl` gives the measure of a complement as $\mu \text{univ} - \mu s$, under measurability and finite-measure assumptions.

Independence is Preserved by Right Complements


```
lemma IndepSet.compl_right
  (hA : MeasurableSet A)
  (hB : MeasurableSet B) :
  IndepSet A B → IndepSet A Bᶜ :=
  by
  intro h
  unfold IndepSet
  unfold IndepSet at h
```

$$A \parallel B \Rightarrow A \parallel B^c$$

Assume $A \parallel B$ i.e. $P(A \cap B) = P(A)P(B)$. We simplify our goal:

$$\begin{aligned}
 P(A \cap B^c) &= P(A \cap (\Omega \setminus B)) \\
 &= P((A \cap \Omega) \setminus (A \cap B))
 \end{aligned}$$





```
-- act on the goal
rw [measure_compl hB
    (measure_ne_top _ _)]
-- use that μ univ = 1
rw [measure_univ]
rw [compl_eq_univ_diff]
-- distributivity of intersection
rw [inter_diff_distrib_left]
rw [inter_univ]
-- introduces need +2x assumptions
rw [measure_diff] -- adds 3x goals
-- continue with goal as normal
-- adds 1x goal
· rw [ENNReal.mul_sub]
  rw [mul_one]
  rw [h]
  -- closes 1x inner goal
  simp
-- close 1st new goal
· simp
-- close 2nd new goal
· measurability
-- close 3rd new goal
· simp
```

$$\begin{aligned}
 P(A \cap B^c) &= P(A \setminus (A \cap B)) \\
 &= P(A) - P(A \cap B) \\
 &= P(A) - P(A)P(B) \\
 &= P(A)(1 - P(B)) \\
 &= P(A)P(B^c)
 \end{aligned}$$

Therefore, $A \parallel B^c$.

We prove that if A is independent of B , then A is also independent of B^c . The strategy is to rewrite the complement $B^c = \Omega \setminus B$, then distribute the intersection with A across that set difference. After this, the measure of $A \cap B^c$ becomes the difference between the measure of $A \cap \Omega$ and the measure of $A \cap B$. The first term simplifies to $\mathbb{P}(A)$, and the second is replaced using the original independence assumption. On the right-hand side, the measure of B^c is rewritten using the complement formula.

Equivalence of independence with either B or B^c

```
lemma IndepSet.compl_right_iff
  (hA : MeasurableSet A)
  (hB : MeasurableSet B) :
  IndepSet A B^c ↔ IndepSet A B := by

  -- divide proof into -> and <-
  constructor
  -- · signals proving first goal
  · intro h -- assumption "->"
    -- apply the A=B lemma from RHS
    rw [← compl_compl B]
    -- apply lemma tailored to sets
    apply IndepSet.compl_right hA
    hB.compl
    -- we reach our goal
    exact h

  · intro h
    -- build lemma: compl_right hA hB
    -- then apply it as function to h
    exact compl_right hA hB h
```

Claim: $A \parallel B^c \Leftrightarrow A \parallel B$

Assume $A \parallel B^c$. Then

$$\begin{aligned}
 &A \parallel B^c \\
 \Leftrightarrow &P(A \cap B^c) = P(A)P(B^c) \\
 \Leftrightarrow &P(A \cap B^c) = P(A)(1 - P(B)) \\
 \Leftrightarrow &P(A) - P(A \cap B) = P(A) - P(A)P(B) \\
 \Leftrightarrow &P(A \cap B) = P(A)P(B) \\
 \Leftrightarrow &A \parallel B
 \end{aligned}$$

Assume $A \parallel B$ and invoke the previous proof.

$A \parallel B \iff A \parallel B^c$. We prove the two implication respectively: the first is obtained from the previous complement theorem, applied directly. For the other, we rewrite B as $(B^c)^c$ and then apply the same result again, now to the set B^c .

Left Complements

```
lemma IndepSet.compl_left
  (hA : MeasurableSet A)
  (hB : MeasurableSet B)
  (h : IndepSet A B) :
  IndepSet Ac B := by
    -- first swap LHS and RHS
    apply IndepSet.symm
    -- apply property proven before
    apply IndepSet.compl_right hB hA
    -- swap back LHS and RHS
    apply IndepSet.symm
    -- we are exactly where we want
    exact h
```

Claim: $A \parallel B \Rightarrow A^c \parallel B$

Assume $A \parallel B$. Then

$$\begin{aligned} A \parallel B &\Leftrightarrow P(A \cap B) = P(A)P(B) \\ &\Leftrightarrow P(B \cap A) = P(B)P(A) \\ &\Leftrightarrow B \parallel A \end{aligned}$$

$$\begin{aligned} B \parallel A &\Rightarrow B \parallel A^c \\ &\Leftrightarrow P(B \cap A^c) = P(B)P(A^c) \end{aligned}$$

Then

$$\begin{aligned} P(B \cap A^c) &= P(B)P(A^c) \\ &\Leftrightarrow P(A^c \cap B) = P(A^c)P(B) \\ &\Leftrightarrow A^c \parallel B \end{aligned}$$

The lemma shows that complements on the left also preserve independence. Rather than proving this from scratch, the proof uses symmetry twice: first swap the roles of A and B , then apply the complement-on-the-right result, and finally swap back.

Independence from Itself

```
lemma indep_self
  (h : IndepSet A A) :
  P A = 0 ∨ P A = 1 := by
    -- expand the definition
    unfold IndepSet at h
    -- apply
    rw [inter_self] at h
    -- swap equality LHS-RHS
    symm at h
    -- apply lemma
    rw [mul_self_eq_self_iff] at h
    -- simplify the expression
    simp at h
    -- conclude
    exact h
```

Claim: $A \parallel A \Rightarrow P(A) = 0 \vee P(A) = 1$

Assume $A \parallel A$. Then

$$\begin{aligned} A \parallel A &\Leftrightarrow P(A \cap A) = P(A)P(A) \\ &\Leftrightarrow P(A) = P(A)P(A) \\ &\Leftrightarrow P(A) = P(A)^2 \\ &\Leftrightarrow P(A)^2 = P(A) \\ &\Leftrightarrow P(A)^2 - P(A) = 0 \\ &\Leftrightarrow P(A)(P(A) - 1) = 0 \\ &\Leftrightarrow P(A) = 0 \vee P(A) - 1 = 0 \\ &\Leftrightarrow P(A) = 0 \vee P(A) = 1 \end{aligned}$$


The lemma identifies the only way a set can be independent from itself. Since $A \cap A = A$, the independence condition reduces to the equation $P(A) = P(A)^2$. That means the probability must be an idempotent element, so it can only be 0 or 1. Conceptually, this says self-independence is possible only in trivial cases.

Bayes' Theorem with Nonzero Denominator

```
theorem bayesTheorem
  (hB : P B ≠ 0) :
  P(A|B) = P A * P(B|A) / P B := by
```

Claim:

$$P(B) \neq 0 \Rightarrow P(A | B) = P(A) \frac{P(B|A)}{P(B)}$$



```
-- divide the proof in 2x: A,
by_cases h : P A = 0
-- proof this goal is trivial
· simp [h]
```

```
-- expand the definition
unfold condProb
-- introduces +2x assumptions
rw [ENNReal.mul_div_cancel h]
-- 1st assumption
· rw [inter_comm]
-- 2nd assumption
· simp
```

We split into cases by $\mathbb{P}(A)$:

$\mathbb{P}(A) = 0$.

$$\mathbb{P}(A \mid B) = \mathbb{P}(A) \frac{\mathbb{P}(B \mid A)}{\mathbb{P}(B)}$$

$\mathbb{P}(A) \neq 0$.

$$\begin{aligned} \mathbb{P}(A) \frac{\mathbb{P}(B \mid A)}{\mathbb{P}(B)} &= \mathbb{P}(A) \frac{\frac{\mathbb{P}(B \cap A)}{\mathbb{P}(A)}}{\mathbb{P}(B)} \\ &= \frac{\mathbb{P}(B \cap A)}{\mathbb{P}(B)} \\ &= \frac{\mathbb{P}(A \cap B)}{\mathbb{P}(B)} \\ &= \mathbb{P}(A \mid B) \end{aligned}$$

Assuming $\mathbb{P}(B) \neq 0$, the proof begins by splitting into the case $\mathbb{P}(A) = 0$ and the case $\mathbb{P}(A) \neq 0$. In the first case, the right-hand side simplifies immediately, achieving the identity. In the nonzero case, the definition of conditional probability is unfolded on both sides. $\mathbb{P}(A)$ can be cancelled using `ENNReal.mul_div_cancel`, and the remaining intersection is rewritten with `inter_comm` so both sides match.

Conditioning on Independent Event

```
lemma condProb_of_indepSet
  (h : IndepSet B A)
  (hB : P B ≠ 0) :
  P(A|B) = P A := by

  unfold condProb
  -- apply multiple substitutions
  rw [h.symm, mul_div_assoc,
  ENNReal.div_self, mul_one]
  -- div_self introduces 2x goals:
  -- prove assumption 1
  · exact hB
  -- prove assumption 2
  · simp
```

Claim: $A \parallel B \Rightarrow \mathbb{P}(A \mid B) = \mathbb{P}(A)$

Assume $A \parallel B$. Then

$$\begin{aligned} \mathbb{P}(A \mid B) &= \frac{\mathbb{P}(A \cap B)}{\mathbb{P}(B)} \\ &= \frac{\mathbb{P}(A)\mathbb{P}(B)}{\mathbb{P}(B)} \\ &= \mathbb{P}(A) \left(\frac{\mathbb{P}(B)}{\mathbb{P}(B)} \right) \\ &= \mathbb{P}(A) * 1 \\ &= \mathbb{P}(A) \end{aligned}$$

Conditioning on an independent event does not change the probability of the event being measured. From the definition of conditional probability, independence allows $\mathbb{P}(A \cap B) = \mathbb{P}(A)\mathbb{P}(B)$. The denominator $\mathbb{P}(B)$ cancels using `ENNReal.div_self`, which is why the nonvanishing hypothesis on $\mathbb{P}(B)$ is needed. After simplification, the expression reduces to $\mathbb{P}(A)$.

11. Additional References

11.1. Sources

Many examples were taken from the “Mathematics in Lean” and “Glimpse of Lean”. They are excellent resources for more in-depth study.

- *Glimpse of Lean*, Patric Massot et al., [Online Repository](#)
- *Mathematics in Lean*, leanprover-community, [Online Book](#)
- *Lean 4 Tactic Cheatsheet*, Floris van Doorn, [Source](#)
- *The Mechanics of Proof*, Heather Macbeth. [Online Book](#)
- *Mechanized Proofs of the vNM Utility Theorem*, Jingyuan Li, [Code](#), [Paper](#)
- *Github Lean Project Template*, leanprover-community, [LeanProject Template](#)

11.2. Complete Projects

- [Analytic Number Theory Exponent Database](#) by Terence Tao et al.
- [Equational Theories](#) by Terence Tao et al.
- [Polynomial Freiman Ruzsa Conjecture](#) by Terence Tao et al.
- [Infinity Cosmos](#) by Emily Riehl et al.
- [ABC Exceptions](#) by Bhavik Mehta et al.
- [Proofs from THE BOOK](#) by Moritz Firsching et al.
- [Soundness of FRI](#) by Bolton Bailey et al.
- [Sphere Packing in 8 Dimensions](#) by Maryna Viazovska et al.
- [Groupoid Model of Homotopy Type Theory](#) by Sina Hazratpour et al.
- [Weil’s Converse Theorem](#) by Chris Birkbeck et al.
- [Dirichlet Nonvanishing](#) by Chris Birkbeck et al.
- [Spectral Theorem](#) by Oliver Butterley and Yoh Tanimoto.
- [Automata Theory](#) by Stefan Hetzl et al.
- [Seymour’s Decomposition Theorem](#) by Ivan Sergeyev et al.
- [NeuralNetworks](#) by Matteo Cipollina.

11.3. Smaller Projects

- *Unit Fractions Density Conjecture*: [git](#), [paper](#), [blueprint](#)
- *Liquid Tensor*: [git](#), [paper](#) and [image](#), [blueprint](#).
- *Arithmetic Progressions - Almost Periodicity*: [home](#), [documentation](#), [git](#)
- *Brauer Group and Galois Cohomology*: [paper](#), [code documentation](#), [blueprint](#)
- *Brownian Motion*: [git](#), [documentation](#), [blueprint](#),
- *Toric Varieties*: [home](#), [paper without lean references](#), [blueprint](#), [upstream-ready](#)
- *Combinatorial Games*: [home](#), [git](#),
- *Forbidden Matrix Theory*: [overview](#), [git](#), [blueprint code](#).
- ∞ -Cosmoi for Lean: [home](#), [blueprint](#), [docs](#)

Related but through a different formalization language: [Busy Beaver Challenge](#).)

